
1) MODELO:


2 DIAGRAMAS DE KRIPKE E SMV:

2.1) Diagramas de Kripke:  
2.1.1) Forno:

2.1.2) Temporizador de 5 min:


2.1.3) Temporizador de 30 min:


2.1.4) Região Crítica:


2.1.5) Semáforo Para Controle de Região Crítica:



2.2) Código em SMV:

--==============================================================================

-- NOME: FORNOS.SMV

--

-- Projeto: Problema dos fornos de uma pizzaria

--

-- Versao: 1.10

-- Autores: Juliana Carpes Imperial

--          Renato Figueiro' Maia

--==============================================================================

--==============================================================================

-- Versao anterior: 1.0 

-- Modificacao: retirada de um estado desnecessario do S.T. que controla a 

--              regiao critica.

--==============================================================================

--==============================================================================

-- Modulo: main

-- Objetivo: controlar os 3 fornos da pizzaria.

--==============================================================================

MODULE main

VAR

-- cada forno e' um processo que vai rodar em paralelo

   Forno1    : process Forno( Forno2.Estado, Forno3.Estado, bSemaforo ) ;

   Forno2    : process Forno( Forno1.Estado, Forno3.Estado, bSemaforo ) ;

   Forno3    : process Forno( Forno1.Estado, Forno2.Estado, bSemaforo ) ;

-- semaforo para controlar area critica

   bSemaforo : boolean                                                  ;

ASSIGN

   init( bSemaforo ) := 1 ; -- area critica esta' disponivel inicialmente

-- testa que os tres fornos nunca vao estar assando ao mesmo tempo

-- ou que vai haver 2 fornos abertos com o outro assando ou aberto

-- (ou seja, os 3 fornos, ao mesmo tempo, ou no estado C ou D),

-- lembrando que no diagrama, para o forno estar aberto, ele deve estar assando 

SPEC

   AG( ( ( Forno1.Estado = Est_C

         & Forno2.Estado = Est_C

         & Forno3.Estado = Est_C )

       | ( Forno1.Estado = Est_C

         & Forno2.Estado = Est_C

         & Forno3.Estado = Est_D )

       | ( Forno1.Estado = Est_C

         & Forno2.Estado = Est_D

         & Forno3.Estado = Est_C )

       | ( Forno1.Estado = Est_C

         & Forno2.Estado = Est_D

         & Forno3.Estado = Est_D )

       | ( Forno1.Estado = Est_D

         & Forno2.Estado = Est_C

         & Forno3.Estado = Est_C )

       | ( Forno1.Estado = Est_D

         & Forno2.Estado = Est_C

         & Forno3.Estado = Est_D )

       | ( Forno1.Estado = Est_D

         & Forno2.Estado = Est_D

         & Forno3.Estado = Est_C )

       | ( Forno1.Estado = Est_D

         & Forno2.Estado = Est_D

         & Forno3.Estado = Est_D ) ) = 0 )

--==============================================================================

-- Modulo: Forno

-- Objetivo: controlar um forno da pizzaria.

--==============================================================================

MODULE Forno( Estado_A, Estado_B, bSemaforo )

VAR

-- Estados do forno de acordo com o diagrama

   Estado    : { Est_A, Est_B, Est_C, Est_D } ;

-- Regioes onde um processo pode estar

   Regiao    : { Fora, Dentro, Entrando }     ;

-- booleano que indica se ja' passaram 5 min

   bPassou5  : boolean                        ;

-- booleano que indica se ja' passaram 30 min

   bPassou30 : boolean                        ;

ASSIGN

   init( Estado ) := Est_A ;    -- Estado inicial e' forno desligado e fechado

   init( Regiao ) := Fora ;     -- Regiao inicial e' fora da area critica

   init( bPassou5 ) := 0 ;      -- inicialmente nao passaram 5 min (desligado)

   init( bPassou30 ) := 0 ;     -- inicialmente nao passaram 30 min (desligado)

-- proximo estado do forno

   next( Estado ) :=

      case

      -- se esta' desligado, fechado e dentro da regiao critica, pode continuar

      -- assim ou ir para pre-aquecimento (continuando fechado)

         ( Estado = Est_A

         & Regiao = Dentro ) :

            { Est_A, Est_B } ;

      -- se esta' pre-aquecendo, fechado, dentro da regiao critica, ja' passaram

      -- 5 min e os outros 2 fornos nao estao assando e/ou abertos, ele vai

      -- assar

         ( Estado = Est_B

         & Regiao = Dentro

         & bPassou5 = 1

         & ( ( Estado_A = Est_C

             & Estado_B = Est_C )

           | ( Estado_A = Est_C

             & Estado_B = Est_D )

           | ( Estado_A = Est_D

             & Estado_B = Est_C )

           | ( Estado_A = Est_D

             & Estado_B = Est_D ) ) = 0 ) :

            { Est_C } ;

      -- se esta' pre-aquecendo, fechado, dentro da regiao critica, ja' passaram

      -- 5 min e os outros 2 fornos estao assando e/ou abertos, ele vai desligar

         ( Estado = Est_B

         & Regiao = Dentro

         & bPassou5 = 1

         & ( ( Estado_A = Est_C

             & Estado_B = Est_C )

           | ( Estado_A = Est_C

             & Estado_B = Est_D )

           | ( Estado_A = Est_D

             & Estado_B = Est_C )

           | ( Estado_A = Est_D

             & Estado_B = Est_D ) ) = 1 ) :

            { Est_A } ;

      -- se esta' assando, fechado, dentro da regiao critica e ja' passaram 30

      -- min, ele vai abrir

         ( Estado = Est_C

         & Regiao = Dentro

         & bPassou30 = 1 ) :

            { Est_D } ;

      -- se esta' assando, aberto e dentro da regiao critica, vai comecar o

      -- ciclo novamente 

         ( Estado = Est_D

         & Regiao = Dentro ) :

            { Est_A } ;

         1 :

            { Estado } ;

      esac ;

-- proximo estado da regiao do processo

   next( Regiao ) :=

      case   

      -- se esta' fora da regiao critica, pode continuar fora ou pode tentar

      -- entrar dentro dela

         Regiao = Fora :

            { Fora, Entrando } ;

      -- se esta' tentando entrar na regiao critica, so' vai faze-lo se o

      -- semaforo estiver aberto

         ( Regiao = Entrando

         & bSemaforo = 1 ) :

            { Dentro } ;

      -- se esta' dentro da regiao critica, so' pode sair se nao estiver com a

      -- pizza pronta

         ( Regiao = Dentro

         & ( ( Estado = Est_C

             & bPassou30 = 1 ) = 0 ) ) :

            { Dentro, Fora } ;

         1 :

            { Regiao } ;

      esac ;

-- mudanca do tempo de pre-aquecimento

   next( bPassou5 ) :=

      case

      -- se ainda nao passaram 5 min, o forno esta' pre-aquecendo, fechado e na

      -- regiao critica, entao, o tempo pode passar ou nao

         ( bPassou5 = 0

         & Estado = Est_B

         & Regiao = Dentro ) :

            { 0, 1 } ;

      -- se ja' passaram 5 min, o forno esta' desligado, fechado e na regiao

      -- critica, entao o tempo vai ser zerado

         ( bPassou5 = 1

         & Estado = Est_A

         & Regiao = Dentro ) :

            { 0 } ;        

         1 :

            { bPassou5 } ;

      esac ;

   next( bPassou30 ) :=

      case

      -- se ainda nao passaram 30 min, o forno esta' assando, fechado e na

      -- regiao critica, entao, o tempo pode passar ou nao

         ( bPassou30 = 0

         & Estado = Est_C

         & Regiao = Dentro ) :

            { 0, 1 } ;

      -- se ja' passaram 30 min, o forno esta' desligado, fechado e na regiao

      -- critica, entao o tempo vai ser zerado

         ( bPassou30 = 1

         & Estado = Est_A

         & Regiao = Dentro ) :

            { 0 } ;        

         1 :

            { bPassou30 } ;

      esac ;

   next( bSemaforo ) :=

      case

      -- se um processo entra na regiao critica, o semaforo e' fechado

         Regiao = Entrando :

            { 0 } ;

      -- se um processo sai da regiao critica, o semaforo e' aberto

         Regiao = Fora :

            { 1 } ;       

      esac ;

-- testa se e' possivel uma pizza ficar queimada 

SPEC

   AG ( ( Estado = Est_C

        & Regiao = Dentro

        & bPassou30 = 1 )

     -> AF( Estado = Est_D ) )

FAIRNESS

   running
2.3) Resultado do SMV:

Ao se aplicar como entrada ao SMV o código que representa o modelo, obteve-se a seguinte saída:

-- specification AG (Estado = Est_C & Regiao = Dentro & b... (in module Forno1) is true

-- specification AG (Estado = Est_C & Regiao = Dentro & b... (in module Forno2) is true

-- specification AG (Estado = Est_C & Regiao = Dentro & b... (in module Forno3) is true

-- specification AG (Forno1.Estado = Est_C & Forno2.Estad... is true

resources used:

processor time: 0.11 s, 

BDD nodes allocated: 5563

Bytes allocated: 1105876

BDD nodes representing transition relation: 498 + 1
3) CCS E WORKBENCH:

3.1) Descrição em CCS:

O estado inicial é com o forno fechado e desligado:

Forno = Est_A

O forno pode ir para pré-aquecimento e continuar fechado ou continuar desligado e fechado:

Est_A = Est_B 

O forno pode continuar em pré-aquecimento e fechado, pode começar a assar e continuar fechado depois de terem passado 5 minutos ou ser desligado e continuar fechado se outros dois fornos assarem antes dele:

Est_B = assar:passou5:Est_C + assar:assar:Est_A


O forno pode continuar assando e fechado ou ser aberto depois de terem passado 30 min:

Est_C = passou30:Est_D


O forno pode continuar aberto e assando ou ser fechado e desligado para assar outra pizza:

Est_D = Est_A


O sistema da pizzaria são os três fornos funcionando em paralelo (ou concorrentemente):

Pizzaria = Forno || Forno || Forno

3.2) CCS Assíncrono:

3.2.1) Código em Workbench:

*===============================================================================

* NOME: FORNOS_CCS.CWB

*

* Projeto: Problema dos fornos de uma pizzaria

*

* Versao: 1.00

* Autores: Juliana Carpes Imperial

*          Renato Figueiró Maia

*===============================================================================

*===============================================================================

* processos que representam o forno

*===============================================================================

* o estado inicial é com o forno fechado e desligado

agent Forno = Est_A ;

* o forno pode ir para pré-aquecimento e continuar fechado ou continuar

* desligado e fechado

agent Est_A = eps.Est_A + Est_B ;

* o forno pode continuar em pré-aquecimento e fechado, pode começar a assar e

* continuar fechado depois de terem passado 5 minutos ou ser desligado e

* continuar fechado se outros dois fornos assarem antes dele

agent Est_B = eps.Est_B + assar.passou5.Est_C + 'assar.'assar.Est_A ;

* o forno pode continuar assando e fechado ou ser aberto depois de terem passado

* 30 min

agent Est_C = eps.Est_C + passou30.Est_D ;

* o forno pode continuar aberto e assando ou ser fechado e desligado para assar

* outra pizza

agent Est_D = eps.Est_D + Est_A ;

* o sistema da pizzaria são os três fornos funcionando concorrentemente com a

* restrição de que a transição assar não ocorra sozinha, para forçar a

* sincronização de transições complementares e não permitir que os três

* fornos assem ao mesmo tempo

agent Pizzaria = ( Forno | Forno | Forno )\{ assar } ;

*===============================================================================

* proposições para teste

*===============================================================================

* é possível uma pizza queimar?

prop Queimar = max( Inv.[ passou30 ]T & [-]Inv ) ;

* é impossível haver a transição assar sem estar sincronizada com a complementar

* a ela (se não houver a sincronização, pode haver sobrecarga na pizzaria)?

prop NaoSobre = max( Inv.[ assar ]F & [-]Inv ) ;

*===============================================================================

* checagem de propriedades utilizando as proposições

*===============================================================================

checkprop( Forno, Queimar ) ;     * testa se a pizza pode queimar

checkprop( Pizzaria, NaoSobre ) ; * testa se pode haver sobrecarga na pizzaria

*===============================================================================

* checagem de propriedades utilizando comandos

*===============================================================================

* vê quais são os estados alcançáveis quando o forno está assando e fechado

* depois que passam 30 min

derivatives( passou30, Est_C ) ;

deadlocks Pizzaria ; * testa se o sistema pode entrar em deadlock

3.2.2) Resultado do Workbench:

true

true

1: Est_D

None.

3.3) CCS Síncrono:
3.3.1) Código em Workbench:

*===============================================================================

* NOME: FORNOS_SCCS.CWB

*

* Projeto: Problema dos fornos de uma pizzaria

*

* Versao: 1.00

* Autores: Juliana Carpes Imperial

*          Renato Figueiró Maia

*===============================================================================

*===============================================================================

* processos que representam o forno

*===============================================================================

* o estado inicial é com o forno fechado e desligado

agent Forno = Est_A ;

* o forno pode ir para pré-aquecimento e continuar fechado ou continuar

* desligado e fechado

agent Est_A = 1:Est_A + Est_B ;

* o forno pode continuar em pré-aquecimento e fechado, pode começar a assar e

* continuar fechado depois de terem passado 5 minutos ou ser desligado e

* continuar fechado se outros dois fornos assarem antes dele

agent Est_B = 1:Est_B + assar:passou5:Est_C + 'assar:'assar:Est_A ;

* o forno pode continuar assando e fechado ou ser aberto depois de terem passado

* 30 min

agent Est_C = 1:Est_C + passou30:Est_D ;

* o forno pode continuar aberto e assando ou ser fechado e desligado para assar

* outra pizza

agent Est_D = 1:Est_D + Est_A ;

* o sistema da pizzaria são os três fornos funcionando em paralelo onde apenas

* as transições em que passaram 5 e 30 min (além de idle e tau) podem ocorrer

* para forçar a sincronização de transições complementares e não permitir que os

* três fornos assem ao mesmo tempo

agent Pizzaria = ( Forno | Forno | Forno )\{ passou5, passou30 } ;

*===============================================================================

* proposições para teste

*===============================================================================

* é possível uma pizza queimar?

prop Queimar = max( Inv.[ passou30 ]T & [-]Inv ) ;

* é impossível haver a transição assar sem estar sincronizada com a complementar

* a ela (se não houver a sincronização, pode haver sobrecarga na pizzaria)?

prop NaoSobre_1 = max( Inv.[ assar ]F & [-]Inv ) ;

prop NaoSobre_2 = max( Inv.[ assar#assar ]F & [-]Inv ) ;

prop NaoSobre_3 = max( Inv.[ assar#assar#assar ]F & [-]Inv ) ;

* a sentença a ser checada é a conjunção das três acima

prop NaoSobre = NaoSobre_1 & NaoSobre_2 & NaoSobre_3 ;

*===============================================================================

* checagem de propriedades utilizando as proposições

*===============================================================================

checkprop( Forno, Queimar ) ;     * testa se a pizza pode queimar

checkprop( Pizzaria, NaoSobre ) ; * testa se pode haver sobrecarga na pizzaria

3.3.2) Resultado do Workbench:

true

true

4) COMPARAÇÕES:

Uma grande vantagem do modelo feito utilizando diagramas de Kripke em relação ao feito em CCS é que no primeiro se enxerga mais facilmente como processos paralelos se sincronizam, já que isso é feito diretamente de acordo com os estados atuais. Além disso, é mais fácil construir o comportamento dos sistemas, já que se pode tomar uma decisão utilizando os estados atuais. 
Sincronização em CCS é muito dificultada com o uso de transições complementares que se sincronizam. Afinal, com elas, não se pode dizer explicitamente quais estados serão alcançados. 

Todavia, utilizando diagramas de Kripke, são necessários vários sistemas de transição auxiliares para fazer a sincronização dos processos principais do sistema. Isso não acontece utilizando CCS, já que com este basta utilizar as transições complementares. Como consequência, tem-se que os sistemas descritos em CCS geralmente ficam muito menores do que os descritos utilizando a abordagem anterior.

Já para checar propriedades, CTL é mais fácil, não só para se construir as sentenças como também para entendê-las, já que utiliza as propriedades dos estados e os caminhos percorridos no S.T. Isso é algo que não acontece com o (-calculus e a lógica de Henessy-Milner, porque é preciso utilizar apenas os rótulos das transições e o ponto fixo junto com invariante do sistema de transição. Porém, as propriedades descritas utilizando a última lógica ficam bem menores do que as da primeira. Essas características ficam bem visíveis nas propriedades utilizadas em cada especificação nos códigos mostrados anteriormente.

Como pôde ser visto, o sistema em CCS foi descrito utilizando o CCS assíncrono e o síncrono no Workbench. No primeiro, apenas uma transição pode ocorrer de cada vez (concorrência), com exceção para as transições complementares, que podem se sincronizar ou não. Já no segundo, todas as transições ocorrem simultaneamente (paralelismo), podendo haver sincronia ou não. No primeiro caso, para haver sincronia, restringiu-se o sistema para que ele não realizasse uma transição sem que ela fosse sincronizada. E no segundo, apenas se permitiu transições silenciosas, idles e as que não têm complementares para o mesmo fim.

Desta forma, conseguiu-se dois sistemas que satisfazem os requisitos, sendo um concorrente e outro paralelo utilizando CCS. 

Já utilizando o modelo de Kripke, os fornos estão funcionando concorrentemente, pois cada um é um processo. Com isso, um forno é escolhido não-deterministicamente para rodar a cada vez durante a simulação do sistema. Este esquema é mais parecido com o do CCS assíncrono. Contudo, como em CCS a sincronia é feita em termos de transições, para se fazer um forno assar, um outro precisa assar e o terceiro precisa desligar. Já no SMV, um (ou dois) forno(s) pode(m) funcionar e os outros (outro) podem (pode) ficar parados (parado). 

